Smoothing data points

Posted on 2010-05-05 07:05 By Christopher2222 (374)
Categories: How do I ...? with Maple



I want to smooth the data but retain the shape. Here is a smaller set of points for an example.

```
with(plots):
a := [[1, 1], [3, 2], [3.5, 4], [4, 6], [5, 5], [6, 7], [7, 6], [8, 5], [9, 5.5], [10, 4], [11, 1], [12, -5], [11.5, -6], [12, -12], [10, -10], [8, -14], [7, -10], [3, -10], [2, -5], [1, -8], [0, 0], [1, 1]];
lisplot(a);
```



I want to use data smoothing to reduce the number of points and still retain the general shape. I suppose I could take out every other pointset but I want an average shape with a minimum number of points. I could probably best fit a circle but that's not what I'm after right now. I don't think best fit line would work as it would just draw a straight line. Any ideas?

‹ error in proc NLPSolve of an integral expression? ›

Login or register to post comments

A moving average sounds like

Comment on 2010-05-05 07:33 from alex_01 (552)
Login or register to post comments

A moving average sounds like the tool you want ?!

```
restart:
randomize():
with(plots):
with(Statistics):

a := 0:  b := 1:  n := 500:  mov := 50:

r := Sample(RandomVariable(Normal(0, 3)), n):

s[1] := 100:   for i from 2 to n do s[i] := a+b*s[i-1]+r[i] end do:

ss := [seq(s[i], i = 1 .. n)]:
tt := [seq(i, i = 1 .. n)]:

A1 := plot(tt, ss, color = black, thickness = 2):
A2 := plot(tt[mov .. n], MovingAverage(ss, mov), color = green, thickness = 3):
display({A1, A2});
```

BSplineCurve

Comment on 2010-05-05 18:36 from alec (2621)
Login or register to post comments



Add [0,0] at the beginning of your data, and [3,2] at the end, and then BSplineCurve may be used,

```
a := [[0, 0], [1, 1], [3, 2], [3.5, 4], [4, 6], [5, 5],
[6, 7], [7, 6], [8, 5], [9, 5.5], [10, 4], [11, 1],
[12, -5], [11.5, -6], [12, -12], [10, -10], [8, -14],
[7, -10], [3, -10], [2, -5], [1, -8], [0, 0], [1, 1],[3, 2]];
```

```
p := CurveFitting:-BSplineCurve(a, x):
plot([p, a], thickness = [2, 1], color = [red, blue]);
```



Alec

Average

Comment on 2010-05-05 09:37 from Christopher2222 (374)
Login or register to post comments

Thanks alex_01 but the BSpline method looks easier.

The MovingAverage method doesn't work on a list of [x,y] data points. You first have to seperate them and then average them and then recombine (there must be a simpler command for that). This is what I came up with before I saw the BSpine way (thanks Alec). Here's what I came up with using the MovingAverage method.

```
with(plots):
with(Statistics):
a := [[1, 1], [3, 2], [3.5, 4], [4, 6], [5, 5], [6, 7], [7, 6], [8, 5], [9, 5.5], [10, 4], [11, 1], [12, -5], [11.5, -6], [12, -12], [10, -10], [8, -14], [7, -10], [3, -10], [2, -5], [1, -8], [0, 0], [1, 1]]:
b:=[seq(op(1,(op(i,a)),i=1..nops(a))]:
c:=[seq(op(2,(op(i,a)),i=1..nops(a))]:
bb:=MovingAverage(b,2):
cc:=MovingAverage(c,2):
dd:=[seq([bb[i],cc[i]],i=1.nops(bb))]:
listplot(dd)
or plot(dd) will work too
```



reduce number of data points

Comment on 2010-05-05 09:41 from Christopher2222 (374)
Login or register to post comments

I still have the same number of data points. How can I reduce that number with minimal loss of basic shape information.

You cant have the same

Comment on 2010-05-05 09:52 from alex_01 (552)

Login or register to post comments

You cant have the same number of data points if you use a moving average:

```
restart:
with(plots):
with(Statistics):

a := [[1, 1], [3, 2], [3.5, 4], [4, 6], [5, 5], [6, 7], [7, 6], [8, 5], [9, 5.5], [10, 4], [11, 1], [12, -5], [11.5, -6], [12, -12], [10, -1], [8, -14], [7, -10], [3, -10], [2, -5], [1, -8], [0, 0], [1, 1]]:

b := [seq(op(1, op(i, a)), i = 1 .. nops(a))]:
bb := MovingAverage(b, 2):
nops(a);
nops(bb);

                              22
                              21


restart:
with(plots):
with(Statistics):

a := [[1, 1], [3, 2], [3.5, 4], [4, 6], [5, 5], [6, 7], [7, 6], [8, 5], [9, 5.5], [10, 4], [11, 1], [12, -5], [11.5, -6], [12, -12], [10, -1], [8, -14], [7, -10], [3, -10], [2, -5], [1, -8], [0, 0], [1, 1]]:

b := [seq(op(1, op(i, a)), i = 1 .. nops(a))]:
bb := MovingAverage(b, 10):
nops(a);
nops(bb);

                              22
                              13
```

moving average window

Comment on 2010-05-05 10:02 from Christopher2222 (374)
Login or register to post comments

Whoops right, thanks. My mistake I forgot about the window size value.

pointloss information lost on endpoints

Comment on 2010-05-05 10:19 from Christopher2222 (374)
Login or register to post comments

However the larger the window size more of the information is lost at the ends of the list. It's not really a reduction in data points the way I thought.

smoothed shape

Comment on 2010-05-05 13:08 from Robert Israel (2772)
Login or register to post comments

You could fit using trig polynomials.

```
a := [[1, 1], [3, 2], [3.5, 4], [4, 6], [5, 5], [6, 7], [7, 6], [8, 5],
[9, 5.5], [10, 4], [11, 1], [12, -5], [11.5, -6], [12, -12], [10, -10],
[8, -14], [7, -10], [3, -10], [2, -5], [1, -8], [0, 0], [1, 1]];
with(Statistics):
n := nops(a);
Xfit, Yfit:= seq(
  Fit(add(A[i]*cos(2*Pi*i*t/(n-1))+B[i]*sin(2*Pi*i*t/(n-1)),i=0..5),<(j$j=1..n)>,map2(op,m,a),t),
    m=1..2);
plots[display](plot(a,colour=blue),plot([Xfit,Yfit,t=0..n-1]));
```



smoothed shape

Comment on 2010-05-05 16:32 from Axel Vogt (1515)
Login or register to post comments

Would you mind do 'expand' your compact solution a bit?

It seems worth to follow it besides Maple's notation ...

expand



Comment on 2010-05-05 17:41 from Robert Israel (2772)
Login or register to post comments

Well, the idea is that we want a parametric curve x = X(t), y = Y(t) with [X(j), Y(j)] approximating a[j] for j from 1 to n. Since a[1] = a[n], it seems reasonable to make X(t) and Y(t) periodic with period n-1. The natural thing to look at when talking about periodic functions is trigonometric polynomials: sum(A[j]*cos(2*Pi*i*j/(n-1)) + B[j]*sin(2*Pi*i*j/(n-1)), j=0..k). I rather arbitrarily took k= 4, and used Fit to get best least-squares fits to both x and y coordinates.

Decreasing the number of points



Comment on 2010-05-05 13:44 from alec (2621)
Login or register to post comments

You can decrease the number of points using one of the "smoothing" curves - in my example with splines, or in Robert Israel's example with trigonometric polynomials.

In my example, if you want to decrease the number of points to 12, that would work as

```
b:=op([1,1],plot(p,numpoints=12,adaptive=false));

  b := [[1.16666666699999988, 1.],
        [3.49870596809588142, 3.99482389413156724],
        [4.73689407180060140, 5.45419849803051982],
        [6.68984183090909212, 6.29523985738135482],
        [8.65728990727272852, 5.24405147212422485],
        [10.6153870463636366, 2.02309127723753246],
        [11.7518227075357054, -5.34373348722703411],
        [11.2423592600872908, -10.9718829544221262],
        [7.80620132550140244, -12.4404205864451463],
        [3.07099554657893358, -8.60860261419190209],
        [0.816478572181008722, -5.53215607104503171],
        [1.16666666999999790, 1.]]
```

Alec

Comment viewing options

Threaded list - expanded | Date - oldest first | 50 comments per page | Save settings

Select your preferred way to display the comments and click "Save settings" to activate your changes.

}